科学・技術の世界

数式処理システムによる新時代の数学

(http://www.math.sci.hokudai.ac.jp/~nami/2017st/)

行木 孝夫、松本 圭司 (北大 理 数学)、
TA 佐藤柾也 (北大 理 数学 修士１年)

第１回, 2017 October 02
次回からは教室を「情報教育館２階 C 室」に変更.

授業の目標:

- 数式処理システム Maxima を用いて，線形代数学や微分積分学における種々の問題を正確に短時間で解けるようにする．
- 数式処理システム Maxima が有するコマンドの理解や多くの計算例を通して，数学科目の理解を深める．

到達目標:

- 数式処理システム Maxima をパソコンにインストールし，その基本的な使い方を修得する．
- 数式処理システム Maxima におけるプログラムを作成および実行が可能になるようにする．
- 数式処理システム Maxima をさまざまな科目の学習に有効利用できるようにする．

Maxima とは何か?

(ウィキペディアより抜粋)
Maxima (マキシマ) は, LISP で記述された数式処理システムである. GNU GPL に基づくフリーソフトウェアであり, 現在も活発に開発が続けられている. Maple や Mathematica などの商用の数式処理システムと比べても遜色のない機能を持っている.

略史:
Maxima の起源は, マサチューセッツ工科大学の MACプロジェクトによって開発され, 米国エネルギー省によって配布されていた Macsyma の1982年のバージョンを GNU Common Lisp に移植したものである.

1982年から Macsyma の独自のバージョンを管理•維持していたBill Schelter (ビル・シェルター (en)) が, 1998年にエネルギー省から GPLライセンスを適用することを条件に公開の許可を得た. こうして公開されたプログラムは Maxima と呼ばれることになり, 2001年のシェルターの死後も開発者や利用者のグループによって独自に開発が続けられている.

Maxima で何ができるのか？

数学科目の演習問題が簡単に解ける.

$\boxed{1}$ 極限 $\displaystyle\lim_{x\to 0}\frac{x\log(1+x)}{1-\cos x}$ を求めよ.

```
limit(x*log(1+x)/(1-cos(x)),x,0);
```

$\boxed{2}$ e^{x^2+x} の Maclaurin 展開の $1, x, \ldots, x^{10}$ の係数を求めよ.

```
taylor(exp(x^2+x), x, 0, 10);
```

$\boxed{3}$ $f(x,y) = x^3 - 3xy + y^3$ の臨界点をすべて求めよ.

```
f:x^3-3*x*y+y^3; solve([diff(f,x)=0,diff(f,y)=0],[x,y]);
```

$\boxed{4}$ 不定積分 $\displaystyle\int (x^2+x+1)e^x(3\sin x + 4\cos x)dx$ を求めよ.

```
factor(integrate((x^2+x+1)*exp(x)*(3*sin(x)+4*cos(x)),x));
```

$\boxed{5}$ 行列 $A = \begin{pmatrix} 2 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ -1 & 2 & -1 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & 2 & -1 & 0 & 0 & 0 & -1 \\ 0 & 0 & -1 & 2 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1 & 2 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & -1 & 2 & -1 & 0 \\ 0 & 0 & 0 & 0 & 0 & -1 & 2 & 0 \\ 0 & 0 & -1 & 0 & 0 & 0 & 0 & 2 \end{pmatrix}$ に対して, $\det(A)$, A^{-1} を求めよ.

```
A:matrix(
[ 2,-1, 0, 0, 0, 0, 0, 0],
[-1, 2,-1, 0, 0, 0, 0, 0],
[ 0,-1, 2,-1, 0, 0, 0,-1],
[ 0, 0,-1, 2,-1, 0, 0, 0],
[ 0, 0, 0,-1, 2,-1, 0, 0],
[ 0, 0, 0, 0,-1, 2,-1, 0],
[ 0, 0, 0, 0, 0,-1, 2, 0],
[ 0, 0,-1, 0, 0, 0, 0, 2]); determinant(A); B:A^^(-1);
```

$\boxed{6}$ 連立方程式 $\begin{pmatrix} 1 & -1 & 1 & 0 \\ 2 & -3 & 5 & -2 \\ -1 & 2 & -4 & 1 \end{pmatrix} \begin{pmatrix} x \\ y \\ z \\ w \end{pmatrix} = \begin{pmatrix} 3 \\ 0 \\ 2 \end{pmatrix}$ を拡大係数行列の簡約化を求めてすべての解を表示せよ.

Maxima には行列の簡約化を求めるコマンドがない. 階段行列まで変形するコマンド echelon を用いて, 簡約化を完成させるコマンドをプログラミングする.

```
coldim(Z):=length(Z[1]);
kanyaku(x):=block([cx,rk,xx,j1,j2,j3],cx:coldim(x),
rk:rank(x),xx:echelon(x), if rk < 2 then return(xx)
else (for j1 from 2 thru rk do
block(for j2 from j1 thru cx do if xx[j1,j2] = 1
then (for j3 thru j1-1 do if xx[j3,j2] # 0
then xx[j3]:xx[j3]-xx[j3,j2]*xx[j1], return(xx)))),
return(xx));
```

```
A:matrix(
[ 1,-1, 1, 0, 3],
[ 2,-3, 5,-2, 0],
[-1, 2,-4, 1, 2]); kanyaku(A);
```

解は $\begin{pmatrix} x \\ y \\ z \\ w \end{pmatrix} = \begin{pmatrix} 7 \\ 4 \\ 0 \\ 1 \end{pmatrix} + t \begin{pmatrix} 2 \\ 3 \\ 1 \\ 0 \end{pmatrix}$ (t は任意の数). それをチェックする.

```
v:matrix([7],[4],[0],[1])+t*matrix([2],[3],[1],[0]);
submatrix(A,5); submatrix(A,1,2,3,4);
factor(submatrix(A,5).v-submatrix(A,1,2,3,4));
```

関数のグラフを描き，関数をよく理解することができる．

$\sin(x)$ の Maclaurin 展開は，

$$\sum_{n=0}^{\infty} \frac{(-1)^n}{(2n+1)!} x^{2n+1} = x - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!} + \frac{x^9}{9!} - \cdots$$

であり，その部分和が項の個数が増えるごとに $\sin(x)$ の近似をよくしていく様子をグラフを描いて理解する．

```
f0(x):=sin(x); f1(x):=x; f2(x):=x-x^3/(3!); f3(x):=f2(x)+x^5/(5!);
f4(x):=f3(x)-x^7/(7!); f5(x):=f4(x)+x^9/(9!);
wxplot2d([f0(x),f1(x),f2(x),f3(x),f4(x),f5(x)], [x,-%pi,%pi]);
```

$z = f(x,y) = x^3 - 3xy + y^3$ のグラフを描き，臨界点で極値かどうか調べる．

```
wxplot3d(y^3-3*x*y+x^3, [x,-0.5,0.5], [y,-0.5,0.5]);
wxplot3d(y^3-3*x*y+x^3, [x,0.5,1.5], [y,0.5,1.5]);
```

私が研究でよく利用する関数

$$\vartheta_{11}(z,\tau) = \sum_{n\in\mathbb{Z}} \exp[\pi\sqrt{-1}\{(n+\frac{1}{2})^2\sqrt{-1}+2(n+\frac{1}{2})(z+\frac{1}{2})\}].$$

ここで，変数 z は複素数を動き関数も複素数値となる．グラフを描くには実４次元空間が必要であるが，関数の実部と虚部のグラフは描くことができる．また，関数の実部と虚部の和のグラフも描くことができる．

```
cfloat(x):=float(realpart(x))+float(imagpart(x))*%i;
Th11(z):=cfloat(
sum(exp(%pi*%i*((n+1/2)^2*%i+2*(n+1/2)*(z+1/2))),
n,-20,20));
wxplot3d(realpart(Th11(x+%i*y)), [x,-2,2], [y,-7/10,7/10],
[grid,50,50]);
wxplot3d(imagpart(Th11(x+%i*y)), [x,-2,2], [y,-7/10,7/10],
[grid,50,50]);
wxplot3d(realpart(Th11(x+%i*y))+imagpart(Th11(x+%i*y)),
[x,-2,2], [y,-7/10,7/10], [grid,50,50]);
```

グラフのきれいな描写に数学的な知識が必要.

例: 半球の描写

原点中心の半径 1 の球面 S の方程式: $x^2 + y^2 + z^2 = 1$.

その上半部分は $z = \sqrt{1 - x^2 - y^2}$ $(x^2 + y^2 \leq 1)$ である. この式から半球を描かせる.

```
wxplot3d(sqrt(1-x^2-y^2), [x,-1,1], [y,-1,1], [grid,60,60]);
```

球面の極座標表示: $(x, y, z) = (\sin\theta\cos\phi, \sin\theta\sin\phi, \cos\theta)$
$(0 \leq \theta \leq \pi, 0 \leq \phi \leq 2\pi)$.

この式から半球を描かせる.

```
wxplot3d([sin(s)*cos(t),sin(s)*sin(t),cos(s)],
[s,0,%pi/2],[t,0,2*%pi],[grid,30,30]);
```

グリッドが 1/4 にもかかわらず, 第 2 方法の方がグラフが滑らかになる.

Maxima を利用することで，古代の著名な数学者にもまさる計算能力を備えることができる.

この講義で修得した技術を今後の学習や研究に利用してもらいたい.

次回からは教室を「情報教育館２階 C 室」に変更．この講義のホームページ:

http://www.math.sci.hokudai.ac.jp/~nami/2017st/

履修に際して，ノートパソコンを講義に持参できることが望ましい.

しっかりとした数学的な知識が必要なので，微分積分学II，線形代数学II を履修している ことが望ましい.

References

[1] 上見練太郎, 勝股脩, 加藤重雄, 久保田幸次, 神保秀一, 山口佳三 共著, 微分 改訂版, 共立出版, 東京, 2014.

[2] 上見練太郎, 勝股脩, 加藤重雄, 久保田幸次, 神保秀一, 山口佳三 共著, 積分 改訂版, 共立出版, 東京, 2014.

[3] 泉屋周一, 石川剛郎, 陳蘊剛, 上見練太郎, 三波篤郎, 西森敏之 共著, 行列と連立一次方程式, 共立出版, 東京, 1996.

[4] 泉屋周一, 石川剛郎, 陳蘊剛, 上見練太郎, 三波篤郎, 西森敏之 共著, 線形写像と固有値, 共立出版, 東京, 1996.

[5] 三宅 敏恒 著,
入門線形代数, 培風館, 東京, 1991.

[6] 三宅 敏恒 著,
入門微分積分, 培風館, 東京, 1992.